ФГБОУ ВО СТАВРОПОЛЬСКИЙ ГОСУДАРСТВЕННЫЙ АГРАРНЫЙ УНИВЕРСИТЕТ
Кафедра «Электротехника, автоматика и метрология»
МЕТОДИЧЕСКИЕ УКАЗАНИЯ

к практическому занятию №9 по дисциплине «Информационные технологии в электроэнергетике»
для студентов направления 13.03.02 – Электроэнергетика и электротехника
Ставрополь
2020
Проекты работы №9 сгруппированы по пяти темам. В каждой теме содержится пять проектов-заданий, но студент выполняет только одно задание, номер которого определяется последним числом в номере зачётки или студбилета следующим образом:

	Последнее число
	Номер задания в теме

	0 или 1
	1

	2 или 3
	2

	4 или 5
	3

	6 или 7
	4

	8 или 9
	5


 1. Движение формы и объектов

1.1. «Прыг-скок»

Описание проекта. После вывода окна проекта путём щелчка по кнопке «Прыг-Скок» на странице Word попытки нажать на кнопку «Нажми меня!» в окне проекта не удаются: кнопка отскакивает в произвольное место окна. Только после 10-й попытки надпись на кнопке меняется на «Сдаюсь!» и нажатие на кнопку закрывает окно проекта. 

Указание. Эффект, демонстрируемый проектом, основан на том, что перескок кнопки CommandButton1, вызываемый событием MouseMove (движение указателя мыши по объекту), происходит раньше, чем событие Click (щелчок) по этой кнопке. Заготовку процедуры-обработчика события MouseMove следует вводить с помощью раскрывающихся списков объектов и событий в верхней части окна кода.

Модификация. Измените код так, чтобы: а) неуспешных попыток нажать кнопку было 15; б) на 15-й попытке надпись на кнопке изменилась на «Устала...»

Private Sub CommandButton1_MouseMove(ByVal Button As _ Integer, ByVal Shift As Integer, ByVal X As Single, ByVal _  Y As Single)

Static Counter As Integer

Randomize

If Counter < 10 Then

 CommandButton1.Top = Rnd * (UserForm1.Height - _

 CommandButton1.Height)

 CommandButton1.Left = Rnd * (UserForm1.Width - _

 CommandButton1.Width)

 Counter = Counter + 1

Else

 CommandButton1.Caption = "Сдаюсь!"

End If

End Sub

Private Sub CommandButton1_Click()

End

End Sub

1.2. «Диск в полёте»

Описание проекта. После вызова окна проекта кнопкой «Летающий DVD» на странице Word и нажатия кнопки «Пуск» в окне объект (изображение DVD-диска) начинает бесконечное движение по форме, отражаясь от её краёв. Кнопка «Пуск» приобретает надпись «Стоп» и служит теперь для останова движения.  

Указание: Проект содержит форму (UserForm1), в которую нужно вставить 2 элемента: кнопку (CommandButton1) и изображение (Image1). Для кнопки устанавливается свойство Caption («Пуск»), а в свойство Picture элемента Image1 вставляется файл ДискDVD.ico из материалов задания. В окне кода вставляется приведенный ниже код. Не забудьте озаглавить форму. Для запуска программы на лист Word поместить запускную кнопку с обработчиком щелчка, визуализирующим форму: UserForm1.Show. Комментарии (слова после знака апострофа вводить не обязательно.

Модификация. Увеличьте скорость по горизонтальной оси в 1.5  раза, а по вертикальной — в 2 раза.

Dim H As Integer, W As Integer, VX As Integer, VY As Integer

Dim T As Single, S As Boolean

Private Sub CommandButton1_Click()

S = Not S 'смена флага выполнения цикла анимации
If S = False Then  'смена надписи на кнопке

 CommandButton1.Caption = "Пуск"

Else

 CommandButton1.Caption = "Стоп"

End If

Do While S = True   ' анимация, если S=True

  Tm = Timer: Do: Loop Until Timer - Tm > 0.01  'псевдо-таймер
  IL = Image1.Left + VX  'новые координаты
  IT = Image1.Top + VY

  If IL < 0 Or IL > W Then   'проверка выхода за пределы поля
    VX = -VX

    IL = W * (IL \ W)   'деление нацело: IL=либо W, либо 0

   ElseIf IT < 0 Or IT > H Then

    VY = -VY

    IT = H * (IT \ H)   'деление нацело: IT=либо H, либо 0

  End If

  Image1.Move IL, IT   'сдвиг изображения
  DoEvents   'позволяет кнопке работать во время анимации

Loop

End Sub

Private Sub UserForm_Activate()

 VX = 2: VY = 2  'скорости по осям
 H = UserForm1.InsideHeight - Image1.Height

  'свободная зона полёта
 W = UserForm1.InsideWidth - Image1.Width

End Sub

Private Sub UserForm_QueryClose(Cancel As Integer, CloseMode As Integer)

 If S = True Then Cancel = True 

 ' окно во время анимации не закрывается

End Sub

1.3. «Вращение формы»

Описание проекта. После щелчка по кнопке «Вращение формы», расположенной на странице Word, сверху экрана появляется небольшое окно с единственной кнопкой и начинает движение против часовой стрелки по спирали, приближаясь к центру. При этом промежуточные изображения остаются, что вызывает узор, показанный на рисунке ниже. В самом центре окно останавливается, «узор» исчезает, и, если нажать кнопку «Закончить», то форма начинает движение в обратном направлении с тем же эффектом «узора» и исчезает за краем экрана.

Указание. Внести форму, в неё вставить кнопку «Закончить» и в окно кода ввести приведёный программный код. Первые две строки с декларацией переменных должны быть действительно первыми в тексте кода. Событие активации формы (UserForm_Activate) происходит после визуализации формы в результате команды UserForm1.Show в процедуре-обработчике щелчка по запускной кнопке «Вращение формы», расположенной непосредственно на странице. 

Модификация. 1. Снимите фиксацию промежуточных положений окна на первой фазе (к центру). Для этого удалите знак апостроф (комментария) у команды DoEvents, т.е., активизируйте её. 2. На второй фазе вращения (от центра) увеличьте в 3 раза шаг движения формы.

Dim PI As Single

Dim X, Y, Nob, RL, RT

Private Sub UserForm_Activate()

PI = 4 * Atn(1)

X = ActiveWindow.Width / 2 - UserForm1.Width / 2

Y = ActiveWindow.Height / 2 - UserForm1.Height / 2

Nob = 5

RL = X / Nob

RT = Y / Nob

UserForm1.Left = X

UserForm1.Top = 0

DoEvents

For R = Nob To 0 Step -0.01

  UserForm1.Left = R * RL * Sin(R * PI * 2) + X

  UserForm1.Top = -R * RT * Cos(R * PI * 2) + Y

  Tm = Timer

  Do:  Loop Until Timer - Tm > 0.01

Next R

End Sub

Private Sub CommandButton1_Click()

 For R = 0 To Nob Step 0.01

  UserForm1.Left = R * RL * Sin(R * PI * 2) + X

  UserForm1.Top = -R * RT * Cos(R * PI * 2) + Y

  Tm = Timer: Do:Loop Until Timer - Tm > 0.01

 Next R

Unload Me

End Sub 

1.4. «Движение формы»

Описание проекта. Нажатие кнопки «Подвижная форма» на странице вызывает появление окна в центре окна приложения (или в центре экрана, если окно развёрнуто на экран). Кнопки со стрелками вызывают гладкое движение формы в соответствующем направлении до края окна приложения. Центральная кнопка немедленно переносит форму в центр окна приложения.  

Указание. Введите код программы, учитывая указанные на иллюстрации имена кнопок. 

Модификация. Добавьте ещё кнопки (всего четыре), которые мгновенно перемещают форму в соответствующий угол окна приложения (или экрана). Код процедур аналогичен коду кнопки  CommandButton5.

Dim T, L

Private Sub UserForm_Activate()

 T = ActiveWindow.Height - UserForm1.Height

 L = ActiveWindow.Width - UserForm1.Width

 UserForm1.Move L / 2, T / 2

End Sub

Private Sub CommandButton1_Click()

For I = UserForm1.Top To 0 Step -0.1

 UserForm1.Move UserForm1.Left, I

 DoEvents

Next I

End Sub

Private Sub CommandButton2_Click()

For I = UserForm1.Left To 0 Step -0.1

 UserForm1.Move I, UserForm1.Top

 DoEvents

Next I

End Sub

Private Sub CommandButton4_Click()

For I = UserForm1.Top To T Step 0.1

 UserForm1.Move UserForm1.Left, I

 DoEvents

Next I

End Sub

Private Sub CommandButton3_Click()

For I = UserForm1.Left To L Step 0.1

 UserForm1.Move I, UserForm1.Top

 DoEvents

Next I

End Sub

Private Sub CommandButton5_Click()

 UserForm1.Move L / 2, T / 2

End Sub

1.5. «Тараканьи бегeq \o (а;´)»

Описание проекта. Имитация старинной забавы. Кнопка «На старт!» перемещает правый конец овалов-«тараканов» на левую границу «бегового» поля за его пределом (позиция старта). Кнопка «Беги!» вызывает движение «тараканов» вправо (к финишу) с переменной скоростью, меняющейся случайно. Кнопка «Замри!» останавливает движение, после чего оно может быть продолжено далее («Беги!») или начато с начала («На старт!»). В правом верхнем углу показывается текущее время «бега» с момента старта в секундах. Когда последний «таракан» приходит к финишу, то выводятся сообщения о победителе, его времени и о величине опережения пришедшего вторым. Кнопки, каждая в свою фазу «состязания», являются временно неактивными.

Указание. Непосредственно на страницу Word поместите, как показано на рисунке, три элементы управления «кнопка». Затем с помощью инструментов рисования Word разместите ниже три автофигуры: «Прямоугольник» (поле забега) и слева от него два «Овала» («тараканы»). Поле забега можно разделить автофигурой «Линия».  Кроме того, поместите под прямоугольником элемент управления «Надпись» (Label1), который будет служить для вывода результата забега, и над прямоугольником справа — надпись Label2, которая будет служить секундомером. Все элементы управления (3 кнопки и 2 надписи) должны как объекты Word иметь формат положения «в тексте». Кнопки «Беги!» и «Стоп» первоначально неактивны (Enabled — False).

Модификация. Ускорьте или замедлите темп движения «тараканов».

Dim V1, V2, H1, H2, W1, W2, Start, Finish, V

Dim TZ As Single

Private Sub Document_Open()

Dim shp As Shape

V1 = 0: V2 = 0: V = 3

Randomize Timer

For Each shp In ActiveDocument.Shapes

I = I + 1

If shp.AutoShapeType = msoShapeRectangle Then R = I

 If shp.AutoShapeType = msoShapeOval Then

  If V1 = 0 Then V1 = I: shp.TextFrame.TextRange.Text = "1" _
  Else V2 = I: shp.TextFrame.TextRange.Text = "2"

 End If

Next

Start = Shapes(R).Left

Finish = Start + Shapes(R).Width

W1 = Shapes(V1).Width

W2 = Shapes(V2).Width

CommandButton1.Enabled = True

CommandButton2.Enabled = False

CommandButton3.Enabled = False

End Sub

Private Sub CommandButton1_Click()

Shapes(V1).Left = Start - W1

Shapes(V2).Left = Start - W2

Label1.Caption = ""

Label2.Caption = "0.000": TZ = 0

CommandButton1.Enabled = False

CommandButton2.Enabled = True

CommandButton3.Enabled = False

End Sub

Private Sub CommandButton2_Click()

CommandButton1.Enabled = False

CommandButton2.Enabled = False

CommandButton3.Enabled = True

H1 = Shapes(V1).Left + W1

H2 = Shapes(V2).Left + W2

FS = False

TS = Timer - TZ

Do

 H1 = H1 + V * Rnd

 H2 = H2 + V * Rnd

 Shapes(V1).Left = H1 - W1

 Shapes(V2).Left = H2 - W2

 T = Timer - TS

 Label2.Caption = Round(T, 3)

  If St1 = False And H1 >= Finish Then T1 = T: St1 = True

  If St2 = False And H2 >= Finish Then T2 = T: St2 = True

  T0 = Timer: Do: DoEvents: Loop Until Timer - T0 > 0.01

  If CommandButton3.Enabled = False Then Exit Sub

Loop Until H1 >= Finish And H2 >= Finish

If T1 < T2 Then N = 1: T = T1 Else N = 2: T = T2

If T1 = T2 Then N = "1+2"

 T = Round(T, 2)

 D = Round(Abs(T1 - T2), 5)

 Fra = "Победил таракан №" & N & " !  "

 Frb = "Время победителя - " & T & " сек., "

 Frc = "опережение - " & D & " сек."

 Label1.Caption = Fra + Frb + Frc

 Label2.Caption = ""

CommandButton1.Enabled = True

CommandButton2.Enabled = False

CommandButton3.Enabled = False

End Sub

Private Sub CommandButton3_Click()

TZ = Label2.Caption

CommandButton1.Enabled = True

CommandButton2.Enabled = True

CommandButton3.Enabled = False

End Sub
2. Работа с текстом на форме

2.1. «Расписание»

Описание проекта. После вывода на лист окна программы (в результате щелчка по находящейся непосредственно на странице Word кнопке «Расписание»), следует выбрать день недели и нажать соответствующую кнопку на форме. В текстовое окно слева выводится список предметов  и вида занятий по ним на данный день недели.

Указание: Проект содержит форму (UserForm1), в которую нужно вставить 7 элементов; 5 кнопок (CommandButton1… CommandButton5), текстовое поле (TextBox1) и Надпись(Label1). Кнопки (справа на форме) и  надпись (сверху) получают свойство Caption, как показано на рисунке, а содержимое свойства Text элемента TextBox1 стирается. Размер шрифта элемента TextBox1 — 12, свойство MultiLine — True, свойство WordWrap — True. В окно кода вставляется нижеприведенный код. Не забудьте озаглавить форму. Для запуска программы на лист Word следует поместить запускную кнопку с обработчиком щелчка, визуализирующим форму: UserForm1.Show.

Модификация. Добавьте кнопку и соответствующий код для дня «Суббота» с предметами «Физическая культура» и «Информатика».
Private Sub CommandButton1_Click()

TextBox1.Text = "История - лекция" + vbCrLf + _

"Философия - практическое занятие" + vbCrLf + _

"Русский язык - семинар" + vbCrLf + "Математика - лекция"

End Sub

Private Sub CommandButton2_Click()

TextBox1.Text = "География - практическое занятие" + vbCrLf + _
"История - семинар" + vbCrLf + _

"Русский язык - практическое занятие" + vbCrLf + _

"Информатика - лабораторная работа"

End Sub

Private Sub CommandButton3_Click()

TextBox1.Text = "Физкультура - практическое занятие" + vbCrLf + _
"История религии - семинар" + vbCrLf + _

"Литература - практическое занятие" + vbCrLf + _

"Английский язык - практическое занятие"

End Sub
Private Sub CommandButton4_Click()

TextBox1.Text = "Математика - практическое занятие" + vbCrLf + _
"История государства Российского - семинар" + vbCrLf + _

"Информатика - практическое занятие" + vbCrLf + _

"Английский язык - практическое занятие"

End Sub

Private Sub CommandButton5_Click()

TextBox1.Text = "Экономика - практическое занятие" + vbCrLf + _
"История - лекция" + vbCrLf + _

"Основы редакторской деятельности - практическое занятие" + _
vbCrLf + "Русский - практическое занятие"

End Sub

2.2. «Счастливая семёрка»

Описание проекта.  Окно программы выводится по щелчку по кнопке «Семерка» на листе Word. Кнопка «Пуск» на форме при каждом нажатии вызывает появление в трёх окошках-надписях случайных целых чисел от 0 до 9. Если, хотя бы в одном окошке появляется цифра «7», то она приобретает красный цвет и справа внизу появляется изображение золотых монет. Если «семёрок» нет, то изображение не появляется (исчезает). 

Указание: Проект содержит форму (UserForm1), в которую нужно вставить 6 элементов; кнопку (CommandButton1), изображение (Image1), три надписи (Label1, Label2, Label3), в которые выводятся цифры (следует подобрать крупный шрифт) и просто надпись Label4, содержащая фразу «Счастливая семёрка». Кнопка получает свойство Caption – «Пуск», как показано на иллюстрации выше. а в свойство Picture элемента Image1 вставляется файл Монеты.WMF из материалов задания. Для элемента Image1 следует обязательно установить значение свойства PictureSizeMode: 1‑frmPictureSizeModeStretch. Только в этом случае рисунок будет вписан в рамку. В окне кода вставляется приведенный ниже код-обработчик щелчка по кнопке CommandButton1. Не забудьте озаглавить форму. Для запуска программы на лист Word следует поместить запускную кнопку с обработчиком щелчка, визуализирующим форму : UserForm1.Show.

Модификация. Измените код, чтобы цифры в окнах, кроме «7» были не чёрного, а синего цвета (vbBlue) . 

Private Sub CommandButton1_Click()

 Dim Win As Boolean

 Dim T As Single

 Win = False

 Image1.Visible = False

 Label1.Caption = Int(Rnd * 10)

 Label2.Caption = Int(Rnd * 10)

 Label3.Caption = Int(Rnd * 10)

   Label1.ForeColor = 0

   Label2.ForeColor = 0

   Label3.ForeColor = 0

If Label1.Caption = 7 Then Win = True: Label1.ForeColor = vbRed

If Label2.Caption = 7 Then Win = True: Label2.ForeColor = vbRed

If Label3.Caption = 7 Then Win = True: Label3.ForeColor = vbRed

If Win = True Then Image1.Visible = True

End Sub

2.3. «Исправить текст»

Описание проекта. Проект реализует исправление текста, набранного c ошибочной раскладкой клавиатуры. После вызова окна программы кнопкой «Исправление текста» в текстовое окно вводится текст с ошибочной раскладкой клавиатуры, что выглядит абракадаброй. После установки направления преобразования («на кириллицу» или «на латиницу», с помощью двух зависимых переключателей) и нажатия кнопки «Исправить» в текстовом окне появляется текст с правильной раскладкой.

Указание: Иллюстрация выше показывает окно программы до и после нажатия кнопки «Исправить». Ниже приведён код программы. При вводе нужно следить за правильностью ввода строковых величин R$ и L$. Символы, составляющие эти величины, нужно вводить просто нажимая последовательно кнопки клавиатуры слева направо построчно, начиная с «йцукен» и «qwerty». Обратите внимание, что символ кавычек вводится с помощью функции Chr(), поскольку сами кавычки окаймляют строковые переменные и константы.

Модификация. Добавьте кнопку, нажатие которой вызовет «стирание» записи в текстовом окне.

Private Sub CommandButton1_Click()

R$="йцукенгшщзхъфывапролджэячсмитьбюёЙЦУКЕНГШЩЗХЪФЫВАПРОЛДЖЭ" _

& "ЯЧСМИТЬБЮЁ" & Chr(34) & "№;:?.,/"

L$ = "qwertyuiop[]asdfghjkl;'zxcvbnm,.`QWERTYUIOP{}ASDFGHJKL:" _

& Chr(34) & "ZXCVBNM<>~@#$^&/?|"

DL = Len(L$): DR = Len(R$): If DL <> DR Then Exit Sub

If OptionButton1.Value = True Then

 For I = 1 To DL

  Znak = Mid(L$, I, 1)

  Ispr = Mid(R$, I, 1)

  TextBox1.Text = Replace(TextBox1.Text, Znak, Ispr)

 Next

Else

 For I = 1 To DR

  Znak = Mid(R$, I, 1)

  Ispr = Mid(L$, I, 1)

  TextBox1.Text = Replace(TextBox1.Text, Znak, Ispr)

 Next

End If

End Sub

2.4. «Эволюции текста»

Описание. Проект на форме, вызываемой кнопкой «Эволюции» на странице. После ввода фразы в верхнее окно (нижнее окно заблокировано) её можно отформатировать тремя шрифтами с помощью зависимых переключателей и дополнительно придать начертание (п/ж, курсив) с помощью двух выключателей (элемент управления ToggleButton) [image: image11.png]Y



, которые выглядят, как кнопки, но обладают способностью «залипать». После нажатия кнопки «вниз» введённый текст с неизменённым форматированием перемещается в нижнее текстовое поле, а верхнее поле блокируется. Текст может быть снова отредактирован, отформатирован и перемещён вверх. 

Указание: Проект содержит два текстовых поля, два выключателя, три зависимых переключателя и две кнопки, одна из которых («вверх») при разработке формы делается скрытой. Также при разработке блокируется нижнее текстовое поле. Для обоих окон свойства MultiLine и WordWrap устанавливаются на True.  

Модификация. Добавьте к проекту выключатель, который при нажатии перекрашивает текст в красный цвет, а при отжатии — возвращает тексту чёрный цвет. Придайте переключателю красный цвет. В программе цвет текста в текстовом поле определяет свойство ForeColor; системные константы для красного и чёрного цвета — vbRed и vbBlack соответственно. 

Private Sub CommandButton1_Click()

  If TextBox1.Text = "" Then Exit Sub

    TextBox2.Locked = False

    TextBox2.Text = TextBox1.Text

    TextBox1.Text = ""

    CommandButton1.Visible = False

    CommandButton2.Visible = True

    TextBox1.Locked = True

End Sub

Private Sub CommandButton2_Click()

  If TextBox2.Text = "" Then Exit Sub

    TextBox1.Locked = False

    TextBox1.Text = TextBox2.Text

    TextBox2.Text = ""

    CommandButton1.Visible = True

    CommandButton2.Visible = False

    TextBox2.Locked = True

End Sub

Private Sub OptionButton1_Click()

 TextBox1.Font = "Arial"

 TextBox2.Font = "Arial"

End Sub

Private Sub OptionButton2_Click()

 TextBox1.Font = "Times New Roman"

 TextBox2.Font = "Times New Roman"

End Sub

Private Sub OptionButton3_Click()

 TextBox1.Font = "Tahoma"

 TextBox2.Font = "Tahoma"

End Sub

Private Sub ToggleButton1_Click()

 If ToggleButton1 = True Then

   TextBox1.Font.Bold = True

   TextBox2.Font.Bold = True

 Else

   TextBox1.Font.Bold = False

   TextBox2.Font.Bold = False

 End If

End Sub

Private Sub ToggleButton2_Click()

 If ToggleButton2 = True Then

   TextBox1.Font.Italic = True

   TextBox2.Font.Italic = True

 Else

   TextBox1.Font.Italic = False

   TextBox2.Font.Italic = False

 End If

End Sub
2.5. «Бегущая строка»

Описание проекта. После ввода текста в окно и нажатия кнопки «Пуск» окно скрывается, на форме появляется тот же текст, который медленно движется вправо. При заходе за правую границу окна «исчезнувший» текст непрерывно появляется слева.

 SHAPE  \* MERGEFORMAT 



  Указание: Помимо текстового поля для ввода фразы и двух кнопок, на форму следует поместить две надписи (Label1 и Label2), установив для них свойства BorderStyle = 0-frmBorderStyleNone (без рамки).
Модификация. Измените в 2 раза (в сторону увеличения или уменьшения) скорость движения строки двумя способами: а) путём изменения размера «скачка» надписей: б) путём изменения времени задержки после очередного «скачка». 

Private Sub CommandButton1_Click()

 If TextBox1.Text = "" Then Exit Sub

If CommandButton1.Caption = "Пуск" Then

 Label1.Caption = TextBox1.Text

 Label2.Caption = TextBox1.Text

 Label1.Left = 0: Label2.Left = Me.Width

 W = Label1.Width

 CommandButton1.Caption = "Стоп"

 TextBox1.Visible = False

 Do

  DoEvents

   Label1.Left = Label1.Left + 0.5

   Label2.Left = Label2.Left + 0.5

   If Label1.Left > W Then Label1.Left = -W

   If Label2.Left > W Then Label2.Left = -W

   tm = Timer

   Do Until Timer - tm > 0.005: Loop

 Loop

Else

 Label1.Caption = "": Label2.Caption = ""

 CommandButton1.Caption = "Пуск"

 TextBox1.Visible = True

 TextBox1.SetFocus

End If

End Sub

Private Sub UserForm_Activate()

Label1.Width = Me.Width

Label2.Width = Me.Width

Label1.Height = TextBox1.Height

Label2.Height = Label1.Height

Label1.Top = (Me.InsideHeight - Label1.Height) / 3

Label2.Top = Label1.Top

End Sub

Private Sub CommandButton2_Click()

 End

End Sub

3. Работа с текстом на странице

3.1. «Ударение»

Описание проекта. Позволяет ставить знак ударения над любым одиночным выделенным в тексте символом. При повторном выделении и нажатии кнопки ударения удаляются со всех букв выделения. 

Устано́вка/сня́тие ударе́ния

Указание: Проекте на листе Word, код помещается непосредственно в модуль ThisDocument. Разместите на первой строке страницы управляющую кнопку (показана выше). В окне свойств кнопки обязательно установите для свойства TakeFocusOnClick значение False. Далее производится двойной щелчок по кнопке и в открывшемся окне кода вводится код-обработчик щелчка. Не допустите ошибку при вводе английского слова «Prompt» (подсказка).
Для проверки работы программы ниже кнопки следует ввести ввести пустую строку и в ней произвольные слова или фразы, выделить нужную букву и нажать кнопку. Таким же образом ударение снимается. 

Модификация. Временно измените код таким образом, чтобы вместо ударения ставились и снимались следующие надстрочные знаки: слабое ударение (ỳ), циркумфлекс (ŷ), тильда (ỹ), черта (ē). Для этого везде в программе код 769 следует заменить соответственно на коды 768, 770, 771, 772.
Private Sub CommandButton1_Click()

 Dim A As Range, T As Range

 Set T = Selection.Range

 If Selection.Type = wdSelectionIP Then

  MsgBox Prompt:="Не выделена буква" 'PROMPT: подсказка (англ.)

 ElseIf Len(T.Characters(1)) = 1 Then

   T.Collapse Direction:=wdCollapseEnd

   T.InsertSymbol CharacterNumber:=769, Unicode:=True

   Selection.Collapse Direction:=wdCollapseEnd

 Else

   For Each A In T.Characters

    If AscW(Right(A, 1)) = 769 Then A = Left(A, 1)

   Next A

 End If

End Sub

3.2. «Таблица умножения»

Описание проекта. На странице Word после заголовка расположены две кнопки. Если нажать кнопку «Построить», то возникает таблица умножения в виде матрицы, как показано ниже. Кнопка «Стереть» удаляет таблицу.
Таблица умножения

Указание: Проект без формы, таблица строится на странице Word, код записывается в модуле ThisDocument. Заголовок «Таблица умножения» вводится в тексте, кнопки также имеют положение «в тексте».

Модификация. В проекте применён стиль таблицы «Простая таблица 2». Изменяя код, примените ещё несколько встроенных стилей таблицы, например, «Сетка таблицы», «Сетка таблицы 5», «Изысканная таблица».  

Private Sub CommandButton1_Click()

 Dim I, J As Integer

 CommandButton2_Click

 Selection.EndKey Unit:=wdStory

 ActiveDocument.Tables.Add Range:=Selection.Range, _

 NumRows:=10, NumColumns:=10

 With ActiveDocument.Tables(1)

   .AutoFitBehavior (wdAutoFitContent)

   .Range.ParagraphFormat.Alignment = wdAlignParagraphRight

   .Style = "Простая таблица 2"

   .ApplyStyleLastRow = False

   .ApplyStyleLastColumn = False

   .Cell(10, 10).Borders(wdBorderLeft).LineStyle = _

    wdLineStyleNone

   .Rows.Alignment = wdAlignRowCenter

  For I = 1 To 10 Step 1

   For J = 1 To 10 Step 1

    .Cell(I, J).Range.Text = I * J

   Next J

  Next I

 End With

End Sub

Private Sub CommandButton2_Click()

 If ActiveDocument.Tables.Count <> 0 Then

  For I = 1 To ActiveDocument.Tables.Count

    ActiveDocument.Tables(I).Delete

  Next

 End If

End Sub
3.3. «Палиндромы»

Описание проекта. Палиндром — слово или фраза, которая при чтении справа налево имеет  тот же порядок букв, как и при обычном чтении — слева направо. (без учёта пробелов и знаков препинания). При нажатии кнопки «Перевернуть» за предварительно выделенным абзацем с палиндромом появляется его «обратное» отображение (см. примеры ниже). Кнопка «Снова» снимает все «переводы».

Выделить строку и нажать кнопку [image: image19.emf] 

Перевернуть


Для снятия палиндромов нажать кнопку [image: image20.emf] 

Снова


А дорогу укажу чужаку у города

Он дивен палиндром и ни морд ни лап не видно 

А роза упала на лапу Азора = АРОЗА УПАЛ АН АЛАПУ АЗОР А

Аргентина манит негра = АРГЕН ТИНАМ АНИТНЕГРА

Артем, а в два метра

Вот семь лет и бюллетенит в тине тел любитель местов

Даст еду деду детсад

Демидова завод и мед
Диван незаразен на вид

Йог Антип спит нагой = ЙОГАН ТИПС ПИТНА ГОЙ

Коттедж ждет ток = КОТ ТЕДЖ ЖДЕТТОК
Мама дала ситро торт и сала дамам = МАМАД АЛАС И ТРОТ ОРТИС АЛАД АМАМ

Нажал кабан на баклажан

Морг у базы за бугром

Но милиция отсыпала на лапы сто яиц и лимон

Торт с кофе не фокстрот = ТОРТСКОФ ЕН ЕФОК С ТРОТ

Уж редко рукою окурок держу

Хан и шаман на машинах

Я иду с мечем судия = ЯИДУС МЕЧЕМ С УДИ Я

Я с уколов еле волокуся

Я уж ревел клевер жуя

Указание: Код проекта помещается непосредственно в модуль ThisDocument. Поместите на страницу Word два абзаца с вышеприведенными фразами (Выделить строку… и Для снятия…) и в конце каждой фразы в тексте вставьте управляющую кнопку. Для каждой кнопки производится двойной щелчок по кнопке и в открывшемся окне кода вводятся нижеприведенные коды-обработчики щелчка. В окне свойств кнопки «Перевернуть» установить свойству TakeFocusOnClick значение False. 

 Проверка работы программы. Ниже управляющих кнопок следует ввести несколько палиндромов (по одному на абзац, "переводы" не вводить). Далее следует целиком выделить один из абзацев, содержащих текст палиндрома и нажать кнопку "Перевернуть". Код-обработчик щелчка по кнопке «Снова» (CommandButton2) «стирает» все слова за знаком равенства во всех строках.

Модификация. Измените код так, чтобы между фразами-перевёртышами стоял не знак равенства,  а символ «тильда»: ~
Private Sub CommandButton1_Click()

Dim T As String

Dim S As String

T = UCase(Selection.Text)

 If Right(T, 1) = Chr(13) Then

  T = Left(T, Len(T) - 1)

  Selection.MoveEnd Count:=-1

 End If

 Selection.Collapse Direction:=wdCollapseEnd

For I = Len(T) To 1 Step -1

 S = S + Mid(T, I, 1)

Next

 Selection.InsertAfter " = " + S

 Selection.Collapse Direction:=wdCollapseEnd

End Sub

Private Sub CommandButton2_Click()

Selection.HomeKey Unit:=wdStory

    With Selection.Find

        .ClearFormatting

        .Replacement.ClearFormatting

        .Text = "^0032=*^0013"

        .Replacement.Text = "^0013"

        .Forward = True

        .Wrap = wdFindStop

        .MatchWildcards = True

        .Execute Replace:=wdReplaceAll

    End With

End Sub

3.4. «Число прописью»

Описание проекта. Если на странице записать цифрами натуральное число до 9999 и нажать кнопку «Прописью», то рядом в скобках появится словесная расшифровка числа на русском языке. Нецелое число округляется до целого; числа меньше 1 или больше 9999 не обрабатываются.

[image: image21.emf] 

Прописью


4561  (четыре тысячи пятьсот шестьдесят один)

Указание: Ниже приводится код обработчика щелчка по кнопке «Прописью» (CommandButton1), располагаемый в модуле «ThisDocument». Этот код реализует словесную расшифровку числа. В окне свойств кнопки присвоить свойству TakeFocusOnClick значение False.
Модификация. Измените вывод результата: без скобок, но заглавными буквами. Для этого следует перед строкой вывода результата преобразовать строковую величину ST в верхний регистр с помощью функции Ucase(), а в самой строке вывода убрать открывающую и закрывающую скобки.

Private Sub CommandButton1_Click()
Dim A1(9)

Dim A2(9)

Dim A3(9)

Dim A4(9)

Dim T As Integer

Dim ST As String

A1(1) = "сто"    'сотни

A1(2) = "двести"

A1(3) = "триста"

A1(4) = "четыреста"

A1(5) = "пятьсот"

A1(6) = "шестьсот"

A1(7) = "семьсот"

A1(8) = "восемьсот"

A1(9) = "девятьсот"

A2(1) = "десять"  'десятки

A2(2) = "двадцать"

A2(3) = "тридцать"

A2(4) = "сорок"

A2(5) = "пятьдесят"

A2(6) = "шестьдесят"

A2(7) = "семьдесят"

A2(8) = "восемьдесят"

A2(9) = "девяносто"

' вторая десятка

A3(1) = "одиннадцать"

A3(2) = "двенадцать"

A3(3) = "тринадцать"

A3(4) = "четырнадцать"

A3(5) = "пятнадцать"

A3(6) = "шестнадцать"

A3(7) = "семнадцать"

A3(8) = "восемнадцать"

A3(9) = "девятнадцать"

A4(1) = "один"  'единицы 

A4(2) = "два"

A4(3) = "три"

A4(4) = "четыре"

A4(5) = "пять"

A4(6) = "шесть"

A4(7) = "семь"

A4(8) = "восемь"

A4(9) = "девять"

T = Val(Selection.Text)

T = Int(T + 0.5)

If T < 1 Or T > 9999 Then _ Exit Sub

If T >= 1000 Then

  I = Int(T / 1000)

Select Case I

 Case 1

  ST = "одна "

 Case 2

  ST = "две "

 Case Else

  ST = A4(I) + " "

End Select

Select Case I

 Case 1

  ST = ST + "тысяча"

 Case 2 To 4

  ST = ST + "тысячи"

 Case Else

  ST = ST + "тысяч"

End Select

  T = T - I * 1000

End If

If T >= 100 Then

  I = Int(T / 100)

  ST = ST + " " + A1(I)
  T = T - I * 100

End If

Select Case T

 Case Is >= 20

   I = Int(T / 10)

   ST = ST + " " + A2(I)

   T = T - I * 10

   ST = ST + " " + A4(T)

 Case 11 To 19

   T = T - 10

   ST = ST + " " + A3(T)

 Case 10

   ST = ST + " " + A2(1)

 Case Else

   ST = ST + " " + A4(T)

End Select

Selection.InsertAfter _ " (" + CStr(ST) + ")"

Selection.Collapse _ Direction:=wdCollapseEnd

End Sub

3.5. «Расчётная таблица»

Описание проекта. На листе Word строится таблица и заполняются ячейки с исходными данными (Рис. 1). При нажатии кнопки «Заполнить» данные производных ячеек вычисляются и выводятся в таблицу (Рис. 2). Нажатие кнопки "Очистить" вызывает стирание данных в этих ячейках.

Анализ выполнения плана поставки

книготорговой продукции за квартал

по универсальному книжному магазину

	Группа товаров
	Плановое
 поступление,
 тыс. руб.
	Поступило, тыс. руб.
	Итого за 
квартал, %

	
	
	Январь
	Февраль
	Март
	тыс. руб.
	% вып.

	Художественная литература
	4230
	1260
	1460
	1540
	
	

	Политическая литература
	1560
	510
	560
	560
	
	

	Педагогическая литература
	1120
	390
	400
	410
	
	

	Прочие товары
	510
	170
	180
	180
	
	

	Итого:
	
	
	
	
	
	


Рис. 1
	Группа товаров
	Плановое
 поступление,
 тыс. руб.
	Поступило, тыс. руб.
	Итого за 
квартал, %

	
	
	Январь
	Февраль
	Март
	тыс. руб.
	% вып.

	Художественная литература
	4230
	1260
	1460
	1540
	4260
	100.7

	Политическая литература
	1560
	510
	560
	560
	1630
	104.5

	Педагогическая литература
	1120
	390
	400
	410
	1200
	107.1

	Прочие товары
	510
	170
	180
	180
	530
	103.9

	Итого:
	7420
	2330
	2600
	2690
	7620
	102.7


Рис. 2
Указание: Введите заголовок таблицы, постройте таблицу Word, как показано на рис.1, произведя, где требуется, объединение ячеек по горизонтали и вертикали. Затем следует заполнить текстовые ячейки таблицы и ввести только первичные числовые исходные данные. Данные в производных ячейках (затенены) вводить не следует. На страницу выше заголовка поместите две управляющие кнопки и в качестве «исходного текста» для процедур щелчка по ним введите приведенный ниже программный код, отдельно для первой (CommandButton1) и второй (CommandButton2) кнопки. 

Модификация и проверка работы программы: Убедитесь, что при изменении данных в первичных ячейках вычисляемые ячейки также изменяют содержание (после нажатия кнопки "Заполнить").

Private Sub CommandButton1_Click()

Dim P(5, 6) As Single

Dim I As Integer, J As Integer

For I = 1 To 4

 For J = 1 To 4

  P(I,J) = Val(ThisDocument.Tables(1).Cell(I+2,J +1).Range.Text)

  Next J

 Next I

 For J = 1 To 4

   For I = 1 To 4

      P(5, J) = P(5, J) + P(I, J)

   Next I

 Next J

  For I = 1 To 5

    P(I, 5) = P(I, 2) + P(I, 3) + P(I, 4)

    P(I, 6) = P(I, 5) / P(I, 1) * 100

  Next I

  For I = 1 To 5

    For J = 1 To 6

        P(I, J) = Int(10 * P(I, J) + 0.5) / 10

    Next J

  Next I

With ThisDocument.Tables(1)

   For J = 1 To 6

    .Cell(7, J + 1).Range.Text = P(5, J)

   Next J

   For I = 1 To 4

       For J = 5 To 6

    .Cell(I + 2, J + 1).Range.Text = P(I, J)

   Next J

  Next I

End With

End Sub

Private Sub CommandButton2_Click()

 With ThisDocument.Tables(1)

   For J = 2 To 7

     .Cell(7,J).Range.Text = ""

   Next J

   For I = 3 To 6

    .Cell(I,6).Range.Text = ""

    .Cell(I,7).Range.Text = ""

   Next I

 End With
End Sub

4. Построение фигур на странице

4.1. «Шар»

Описание проекта. Щелчок по кнопке «Шар» вызывает процесс построения на странице Word изображения шара из последовательно рисуемых овалов. Так как овалы в нижней части шара идут чаще, то создаётся впечатление, что шар наполовину заполнен жидкостью. По завершении построения овалы шара группируются в одну фигуру, она приобретает золотистый цвет, а фон — бардовый. Повторное нажатие кнопки удаляет изображение шара, снимает фон страницы и воспроизводит построение с начала.

Указание. В проекте нет формы, код проекта помещается непосредственно в модуль «ThisDocument». Поместить на страницу Word управляющую кнопку «Шар», произвести двойной щелчок по кнопке и в открывшемся окне кода ввести нижеприведенный код-обработчик.

Модификация. Измените программу так, чтобы шар был синего цвета (RGB(0, 0, 255)). Путём изменения в программе параметров T, W, R изучите их роль в построении изображения.
Dim W, R, X, Y   'должна быть первой строкой в программе  

Private Sub CommandButton1_Click()

T = 320

W = 400

R = 0.2

X = PageSetup.PageWidth / 2

Y = T + W * R / 2

If Shapes.Count > 0 Then Range.ShapeRange.Delete

Background.Fill.Visible = msoFalse

 Call Oval(7, 1)

 Call Oval(3, -1)

StoryRanges(wdMainTextStory).ShapeRange.Group

Shapes(1).Line.ForeColor.RGB = RGB(255, 255, 0)

Background.Fill.ForeColor.RGB = RGB(128, 0, 0)

Background.Fill.Visible = msoTrue

Selection.HomeKey

End Sub

Private Sub Oval(S, Z)

    W2 = W * W / 4

 For I = 0 To W / 2 Step S

    WD = 2 * Sqr(W2 - I * I)

    HT = WD * R

    L = X - WD / 2

    T = Y - HT / 2 - Z * I

    Shapes.AddShape(msoShapeOval, L, T, WD, HT).Select

    Selection.ShapeRange.Fill.Visible = msoFalse

 Next

End Sub

4.2. «Конус»

Описание проекта. При нажатии на кнопку с надписью «Конус» на странице Word происходит построение конуса из последовательно выводимых снизу вверх овалов. После построения овалы группируются в одну фигуру, затем фигуре придаётся тень.

Указание. Поместить на страницу Word управляющую кнопку «Конус», которая как объект должна иметь положение «в тексте». Код-обработчик щелчка по кнопке, помещаемый в модуль ThisDocument,  приводится ниже. 

Модификация. Если снять апострофы (символы комментария) с 4, 5 и 6-ой строк с конца программы, то фигура приобретает заливку синего цвета, а тень — красного. Испытайте другие цвета и типы тени.

Private Sub CommandButton1_Click()

If Shapes.Count > 0 Then Range.ShapeRange.Delete

Left1 = 300

Top1 = 250

Width1 = 150

W = Width1

T = Top1

r = 0.5

For I = 1 To Width1 / 2

    L = Left1 - W / 2

    H = W * r

    Shapes.AddShape(msoShapeOval, L, T, W, H).Select

    Range.ShapeRange.WrapFormat.Type = wdWrapThrough

    Shapes(I).Fill.Visible = False

    T = T - 2

    W = W - 2

Next

With Range.ShapeRange

 .Group

 .WrapFormat.Type = wdWrapThrough

 .Fill.Visible = False

 .Shadow.Type = msoShadow6

 '.Shadow.ForeColor = wdColorRed

 '.Fill.ForeColor = wdColorBlue

 '.Fill.Visible = msoTrue

End With

Selection.Collapse

End Sub

4.3. «Ковёр»

Описание проекта. При щелчке по кнопке «Ковёр» на странице Word начинается медленный процесс «плетения» ковра с чередующимися красными и синими полосами. После построения «ковра», он поворачивается на 45°.

Указание:  Кнопка запуска должна иметь положение «в тексте», размещаться на первой строке с выравниванием «влево».

Модификация. Измените цвета «пряжи», изменяя параметр RGB(красный, зелёный, синий). Измените конфигурацию петель, немного изменяя высоту и ширину овалов в команде .Shapes.AddShape msoShapeOval, X, Y, 17, 15
Private Sub CommandButton1_Click()

If Shapes.Count > 1 Then Range.ShapeRange.Delete

 Application.ScreenRefresh

 For Y = 100 To 355 Step 8

  For X = 100 To 355 Step 8

    T = T + 1

    If Int(T / 4) = T / 4 Then

     Shapes.AddShape msoShapeOval, X, Y, 17, 15

     Shapes(T).Line.ForeColor = RGB(0, 0, 128)

    Else

     Shapes.AddShape msoShapeOval, X, Y, 15, 17

     Shapes(T).Line.ForeColor = RGB(128, 0, 0)

    End If

    Shapes(T).Fill.Visible = False

If Int(T / 128) = T / 128 Then Application.ScreenRefresh

  Next X

 Next Y

With Range.ShapeRange

 .Group

 .IncrementRotation 45

 .Fill.Visible = False

End With

End Sub

4.4. «Золотые CD-диски»

Описание проекта. При щелчке по кнопке «Золотые диски» на странице Word на синем фоне появляются два диска золотистого цвета. С помощью переключателей построение дисков может быть быстрым или медленным. 

        [image: image29.wmf] 

[image: image31.wmf] 

Медленно

Указание. Построение фигур производится непосредственно на листе Word. В процедуру-обработчик щелчка по кнопке «Золотые диски» (CommandButton1) в модуле ThisDocument ввести нижеприведённый код. В качестве первой строки следует ввести декларацию переменных. Затем, также в модуль ThisDocument вводится текст подпрограммы Shape(), создающей диски из автофигур (кругов и овалов). В процедуре предусмотрены несколько различающиеся способы вывода автофигур, способ определяется с помощью переключателей OptionButton1 и OptionButton2, размещаемых на странице правее кнопки запуска приложения.

Модификация. Измените наклон диска, расположенного слева, и/или диска, расположенного справа.

Dim stp, W, W1, L, T

Private Sub CommandButton1_Click()

If Shapes.Count > 0 Then

  Range.ShapeRange.Delete

  Background.Fill.Visible = msoFalse

  Exit Sub

End If

PW = PageSetup.PageWidth

T = PW / 3

L = PW / 3

W = PW / 3 'диам. диска

W1 = PW / 4 'толщ. внутр. части диска

W2 = PW / 5 'половина расст. между центрами дисков

stp = 1

Call Shape(1, L - W2)

Call Shape(0.4, L + W2)

StoryRanges(wdMainTextStory).ShapeRange.Group

Shapes(1).Line.ForeColor.RGB = RGB(255, 255, 0)  ' жёлтый
Shapes(1).Fill.Visible = msoFalse

Background.Fill.ForeColor.RGB = RGB(0, 0, 128)  'т.-синий
Background.Fill.Visible = msoTrue

End Sub

Private Sub Shape(R, L)

 For I = 0 To W1 Step stp

    WD = W - I

    HT = WD * R

    X = L + I / 2

    Y = T - HT / 2

  If OptionButton1.Value = True Then

     Shapes.AddShape msoShapeOval, X, Y, WD, HT

   Else

    Shapes.AddShape(msoShapeOval, X, Y, WD, HT).Select

  End If

 Next

End Sub

4.5. «Объём цилиндра»

Описание проекта. На странице Word целиком расположено расчётное средство для определения объёма цилиндра по высоте и радиусу основания. В текстовые окна вводятся эти величины и после нажатия кнопки «Вычислить» снизу появляется ответ. Кнопка «Снова» очищает элементы с данными и подготавливает следующий счёт.

Указание: Текст программы записывается в модуле ThisDocument в виде процедур обработки щелчков по расположенным на странице кнопкам.  Заголовок можно ввести просто как текст или в элементе управления «Надпись». Формула вставляется как объект Microsoft Equation 3.0. Для ввода данных служат два текстовых поля с поясняющими надписями, представляющими собой элементы управления «Надпись». Кнопка «Вычислить» должна иметь имя CommandButton1, а кнопка «Снова» — CommandButton2. Результат выводится в надпись с именем Label1. Размер страницы документа следует подогнать под место, занимаемое инструментами программы. Также можно оформить границу и фон страницы.

Модификация.  Измените проект так, чтобы в нём вычислялась площадь боковой поверхности цилиндра по формуле 
[image: image34.wmf].

 Измените соответствующим образом текст программы и надписи на листе.   
Private Sub CommandButton1_Click()

  Dim H As Single, R As Single, V As Double

  H = Val(TextBox1.Text)

  R = Val(TextBox2.Text)

  If H = 0 Then TextBox1.Text = "": Exit Sub

  If R = 0 Then TextBox2.Text = "": Exit Sub

  pi = 4 * Atn(1)

  V = pi * H * R * R

  V = Round(V, 2)

  Label1.Caption = "Объем цилиндра равен " + CStr(V)

End Sub

Private Sub CommandButton2_Click()

Label1.Caption = ""

TextBox1.Text = ""

TextBox2.Text = ""

End Sub

5. Вычисления

5.1. «Экзамены»

Описание проекта. Вводятся данные по общему числу экзаменов (TextBox1) и вероятности сдать каждый из них (TextBox2). Данные по вероятности можно ввести непосредственно в окно TextBox2 или назначить с помощью полосы прокрутки. При щелчке по кнопке «P =» выводится вероятность того, что хотя бы на одном экзамене студент получит «неуд». Идею проекта предложил проф. Рудяк Ю.В.

 SHAPE  \* MERGEFORMAT 



Указание. При разработке формы учесть, что надписи Label1... Label4 служат пояснениями, а Label5 — для вывода результата. Объект ScrollBar1 — это полоса прокрутки, имеющая горизонтальную ориентацию (свойство Orientation). Установить для объекта TextBox2 свойство Text на 0.995; для объекта ScrollBar1 (полоса прокрутки) свойство Min установить на 500, свойство Max — на 1000, Value — на 995. Текст в окнах должен быть центрирован, а в надписи Label5 — выровнен по левому краю (свойство TextAlign). 
Модификация. Сделайте округление вероятности до 0.1%. Измените допустимую минимальную вероятность сдать отдельный экзамен с 0.5 на 0.7.  

Private Sub CommandButton1_Click()

Ex = Val(TextBox1.Text)

Pr = Val(TextBox2.Text)

If Ex = 0 Then

 MsgBox prompt:="Введите число экзаменов!"

Else

 P = 1 - Pr ^ Ex

 P = Int(P * 10000) / 100 & "%"

 Label5.Caption = CStr(P)

End If

End Sub

Private Sub ScrollBar1_Change()

 SBar = CStr(ScrollBar1.Value / 1000)

 TextBox2.Text = Replace(SBar, ",", ".")

End Sub

Private Sub TextBox2_Change()

 If Val(TextBox2.Text) < 0.5 Or Val(TextBox2.Text) > 1 Then
  ScrollBar1_Change

 Else

  ScrollBar1.Value = Val(TextBox2.Value) * 1000

 End If

End Sub

5.2. «Простые числа»

Описание. Если в два непосредственно на листе Word расположенных текстовых окна ввести диапазон натуральных чисел (второе число может быть больше или меньше первого) и нажать кнопку «Вычислить», то в большое текстовое поле ниже будут выведены через запятую все простые  числа из диапазона. 

3451780081, 3451780093, 3451780097, 3451780153, 3451780159, 

3451780207, 3451780231, 3451780243, 3451780261, 3451780271, 

3451780273, 3451780337, 3451780379, 3451780469, 3451780471, 

3451780523, 3451780543, 3451780601, 3451780639, 3451780679, 

3451780687, 3451780697, 3451780723, 3451780763, 3451780769, 

3451780811, 3451780823, 3451780847, 3451780903, 3451780907, 

3451780919, 3451780921, 3451780933, 3451780951, 3451780957, 

3451780993, 3451781101, 3451781111, 3451781117, 3451781131, 

3451781141, 3451781219, 3451781279, 3451781293, 3451781309, 

3451781339, 3451781363, 3451781413, 3451781449, 3451781453, 

3451781479, 3451781483, 3451781491, 3451781503, 3451781519, 

3451781611, 3451781621, 3451781651, 3451781701, 3451781743, 

3451781761, 3451781783, 3451781789, 3451781797, 3451781873, 

3451781887, 3451781899, 3451781903, 3451781953, 3451782029, 

3451782049, 3451782053, 3451782067, 3451782071, 3451782077, 

3451782083, 3451782091, 3451782157, 3451782233, 3451782253, 

3451782263, 3451782277, 3451782283, 3451782287, 3451782317, 

3451782331, 3451782343, 3451782349, 3451782359, 3451782409, 

3451782491, 3451782511, 3451782547, 3451782557, 3451782617, 


Указание: Предусмотрите вертикальную полосу прокрутки для большого текстового поля (TextBox3). Также установите для него следующие значения свойств: MaltiLine=True; WordWrap=True; SpecialEffect = 2-SpecialEffectSunken.

Модификация. Вставьте дополнительно кнопку, очищающую одновременно все три текстовых окна.
Private Sub CommandButton1_Click()

Dim I As Double, X As Double, Y As Double, Z As Double

TextBox3.Text = ""

X1 = Val(TextBox1.Text)

X2 = Val(TextBox2.Text)

If X2 < X1 Then

 Z = X2: X2 = X1: X1=Z

 TextBox1.Text = Cstr(X1)

 TextBox2.Text = Cstr(X2)

End if

If X1 < 1 Or X2 < 1 Then Exit Sub

If X1 = 1 Then Text3.Text = "1, "

X1 = Int(X1): X2 = Int(X2)

For I = X1 To X2

 Z = Int(Sqr(I))

  For J = 2 To Z

   Y = I / J

   If Int(Y) = Y Then Exit For

  Next

  If J = Z + 1 Then TextBox3 = TextBox3 + CStr(I) + ", "

Next
End Sub

5.3. «Простые сомножители»

Описание. Проект на форме, поэтому на листе Word помещена запускная кнопка проекта. После ввода в текстовое поле натурального числа и нажатия кнопки «Разложить на простые сомножители», во втором текстовом окне в столбик выводятся простые сомножители введённого числа в порядке их возрастания. Снизу для проверки появляется надпись с произведением простых сомножителей.

Указание: Для соответствия текста программы и разработанной формы надпись, в которую выводится произведение, должна иметь имя Label3. Во втором текстовом поле предусмотрите вертикальную полосу прокрутки. Также установите для него следующие значения свойств:

MaltiLine — True; SpecialEffect — 2-SpecialEffectSunken; WordWrap — True.
Модификация. Вставьте дополнительно кнопку, ощищающую все два текстовых окна, а также надпись Label3.
Private Sub CommandButton1_Click()

Dim I As Variant, X As Variant, Y As Variant, Z As Variant
TextBox2.Text = ""

Label3.Caption = ""

UserForm1.Repaint

X = CDec(TextBox1.Text)

If X < 2 Or InStr(CStr(X), ".") <> 0 Then Exit Sub

I = CDec(2)

Z = CDec(1)

Do While X > 1

  If I * I > X Then I = X

  Y = CDec(X / I)

  If (InStr(CStr(Y), ".") = 0 And InStr(CStr(Y), ",") = 0) Then
   TextBox2.Text = TextBox2.Text + CStr(I) + vbCrLf

   UserForm1.Repaint

   Z = Z * I

   X = Y

  Else

   I = I + 1

  End If

Loop

Label3.Caption = "Произведение = " & Z

End Sub

5.4. «Наибольший делитель»

Описание проекта. Кнопка на странице Word  визуализирует окно (форму), на которой в текстовое поле с надписью «Целое число» вводится достаточно большое натуральное число. После нажатия кнопки «Расчёт» в надпись рядом с текстовым полем выводится набольший делитель введённого числа.

Указание: Левое окно (для ввода) — это текстовое поле (TextBox1), а справа (для вывода) — это надпись (Label1). Кроме того, в показанном на иллюстрации месте поместите надпись с именем Label3 (или другим именем с соответствующим изменением в тексте программы). Надпись должна быть без рамки и первоначально невидима. Если введённое число окажется простым, то надпись становится видимой.

Модификация. Вставьте дополнительно кнопку, очищающую одновременно текстовое поле, надпись для вывода (Label1), а также надпись Label3.
Private Sub CommandButton1_Click()

If TextBox1.Text = "" Then Exit Sub

Label1.Caption = ""

Label3.Visible = False

N = Val(TextBox1.Text)

For I = N - 1 To 2 Step -1

 J = N / I

 If J = Int(J) Then Label1.Caption = CStr(I): Exit For

Next

If I = 1 Then

 Label1.Caption = CStr(N)

 Label3.Visible = True

End If

End Sub

5.5. «Наименьший делитель»

Описание проекта. Кнопка на странице Word  визуализирует окно (форму), на которой в текстовое поле с надписью «Целое число» вводится достаточно большое натуральное число. После нажатия кнопки «Расчёт» в надпись рядом с текстовым полем выводится наименьший делитель введённого числа. Чтобы найти число с достаточно большим наименьшим делителем, оно должно заканчиваться на цифры «1» или «7».

Указание: Левое окно (для ввода) — это текстовое поле (TextBox1), а справа (для вывода) — это надпись (Label1). Кроме того, в показанном на иллюстрации месте поместите надпись с именем Label3 (или другим именем с соответствующим изменением в тексте программы). Надпись должна быть без рамки и первоначально невидима. Если введённое число окажется простым, то надпись становится видимой.

Модификация. Вставьте дополнительно кнопку, очищающую одновременно текстовое поле, надпись для вывода (Label1) и надпись Label3.
Private Sub CommandButton1_Click()

If TextBox1.Text = "" Then Exit Sub

Label1.Caption = ""

Label3.Visible = False

N = Val(TextBox1.Text)

For I = 2 To N - 1

 J = N / I

 If J = Int(J) Then Label1.Caption = CStr(I): Exit For

Next

If I = N Then

 Label1.Caption = CStr(N)

 Label3.Visible = True

End If

End Sub

